

# Programming Orangutans and the 3pi Robot from the Arduino Environment




1. Introduction . . . . .	2
2. ATmega168/328-Arduino Pin Mapping . . . . .	5
2.a. Orangutan SV-xx8 and LV-168 Pin Mappings . . . . .	6
2.b. Baby Orangutan B Pin Mappings . . . . .	11
2.c. 3pi Robot Pin Mappings . . . . .	15
3. Configuring the Arduino Environment . . . . .	22
4. Using the Arduino IDE: Blinking an LED . . . . .	24
5. Arduino Libraries for the Orangutan and 3pi Robot . . . . .	28
5.a. OrangutanAnalog - Analog Input Library . . . . .	30
5.b. OrangutanBuzzer - High-Level Buzzer Control Library . . . . .	35
5.c. OrangutanLCD - LCD Control Library . . . . .	46
5.d. OrangutanLEDs - LED Control Library . . . . .	52
5.e. OrangutanMotors - Motor Control Library . . . . .	55
5.f. OrangutanPushbuttons - Pushbutton Interface Library . . . . .	60
5.g. Pololu3pi - Sensor Library for the 3pi Robot . . . . .	63

## 1. Introduction





The Orangutan LV-168 attached to a 3-AA battery holder.



Baby Orangutan B with included header pins soldered in for breadboard installation.

**Arduino** [<http://www.arduino.cc/>] is a popular, open-source prototyping platform that makes it easy for people with little electronics experience to get into the world of microcontrollers and embedded programming. The Arduino environment consists of an open-source integrated development environment (IDE) and compiler, software libraries, and programmable hardware boards. This document explains how to program our **Orangutan SV-xx8** [<https://www.pololu.com/product/1227>], **Orangutan LV-168** [<https://www.pololu.com/product/775>], **Baby Orangutan B** [<https://www.pololu.com/product/1220>], and **3pi robot** [<https://www.pololu.com/product/975>] from the Arduino IDE. The Orangutans and the 3pi robot have a substantial degree of overlap with Arduinos because they use the same ATmega328P and ATmega168 processors found on **official Arduinos** [<https://www.pololu.com/category/125/arduino-compatible>]. Our Orangutan robot controllers can therefore be an attractive alternative to official Arduino hardware for those already familiar with the Arduino environment. Also, the Arduino IDE can be a beginner-friendly and cross-platform alternative to Microchip Studio for those looking to get started with Orangutans.



The Orangutans and 3pi robot offer integrated hardware that Arduinos do not, such as on-board dual motor drivers. This makes them well suited for applications involving control of small- to medium-sized robots. The Orangutan SV-xx8, Orangutan LV-168, and 3pi robot also have integrated pushbuttons, a piezo buzzer, and a removable LCD, all of which are additional features not found on official Arduinos.

With a simple installation of some files in your sketchbook directory, it becomes possible to program our Orangutans and 3pi robot using the Arduino IDE and libraries. This guide will step you through the process of reconfiguring the Arduino IDE, and it will provide a series of custom libraries that will make it easy for you to interface with all of the Orangutan's/3pi's onboard hardware.

One key difference between standard Arduino boards and the Orangutan robot controllers is that most Arduinos come with integrated USB-to-serial adapters, which allow you to connect them to your computer and program them through pre-loaded bootloaders, while Orangutans lack integrated USB-to-serial adapters. As such, for the Orangutan SV-xx8, LV-168 and 3pi robot, we recommend you program the Orangutans or 3pi robot using an AVR ISP (referred to as ICSP in the Arduino world) programmer such as our **USB AVR Programmer** [<https://www.pololu.com/product/1300>]. By taking this approach and leaving off the bootloader, the AVR's hardware serial port is free for other uses, and you gain an extra 512 bytes of program space. You also avoid the bootloader delay when you power up or reset the board.



**Pololu USB AVR programmer with included six-pin ISP cable.**



**Note:** This document applies only to the **3pi robot** [<https://www.pololu.com/product/975>] and **Orangutan SV-xx8** [<https://www.pololu.com/product/1227>], **Orangutan LV-168** [<https://www.pololu.com/product/775>], and **Baby Orangutan B** [<https://www.pololu.com/product/1220>] robot controllers. For simplicity, the latter three devices will be referred to from this point on as “**Orangutans**”, unless a distinction needs to be made between specific Orangutan models. This document does **not** apply to the original **Orangutan** [<https://www.pololu.com/product/225>], **Baby Orangutan** [<https://www.pololu.com/product/215>], **Orangutan X2** [<https://www.pololu.com/product/738>], or **Orangutan SVP** [<https://www.pololu.com/product/1325>] robot controllers.


## 2. ATmega168/328-Arduino Pin Mapping

In our standard Orangutan and 3pi documentation, we use the same convention that Atmel uses in its ATmega168/328 datasheet, referring to pins by their associated ports. For example, pin PD7 is pin 7 of port D. This convention is used because each port is controlled by its own AVR registers, and the bits of these registers correspond to individual pins.

The Arduino insulates the user from this level of detail, referring to the I/O pins as digital pins 0 to 13 and analog inputs 0 to 7. Note that this terminology can be a bit misleading since analog inputs 0 to 5 can also be used as general purpose digital I/O pins (referred to as digital pins 14 to 19).



You can visit <http://www.arduino.cc/en/Hacking/PinMapping168> [http://www.arduino.cc/en/Hacking/PinMapping168] to see exactly how the mega168/328 pins are mapped to Arduino pins. Note that this page shows a 28-pin DIP package chip while the Orangutans carry smaller 32-pin MLF packages that contain two additional dedicated analog inputs: ADC6 and ADC7. In the Arduino environment, these pins become arduino analog inputs 6 and 7, respectively. The following diagram comes from the page linked above:


### Atmega168 Pin Mapping



Digital Pins 11, 12 & 13 are used by the ICSP header for MISO, MOSI, SCK connections (Atmega168 pins 17, 18 & 19). Avoid low-impedance loads on these pins when using the ICSP header.

## 2.a. Orangutan SV-xx8 and LV-168 Pin Mappings





## Pin Assignment Table Sorted by Function


Function	Arduino Pin	mega168 Pin
digital I/Os (x8)	digital pins 0 and 1 analog inputs 0 – 5	PD0, PD1, PC0 – PC5
analog inputs (x8)	analog inputs 0 – 7	PC0 – PC5, ADC6, ADC7
motor 1 control (A and B)	digital pins 5 and 6	PD5 and PD6
motor 2 control (A and B)	digital pins 3 and 11	PD3 and PB3
red user LED	digital pin 1	PD1
green user LED	digital pin 7	PD7
user pushbuttons (x3)	digital inputs 9, 12, and 13	PB1, PB4, and PB5
buzzer	digital pin 10	PB2
LCD control (RS, R/W, E)	digital pins 2, 8, and 4	PD2, PB0, and PD4
LCD data (4-bit: DB4 – DB7)	digital pins 9, 12, 13, and 7	PB1, PB4, PB5, and PD7
user trimmer potentiometer	analog input 7 (through jumper)	ADC7
temperature sensor (LV-168 only) battery voltage monitor (SV-168 only)	analog input 6 (through SMT jumper)	ADC6
ICSP programming lines (x3)	digital pins 11, 12, and 13	PB3, PB4, PB5
reset pushbutton	reset	PC6
UART (RX and TX)	digital pins 0 and 1	PD0 and PD1
I2C/TWI (SDA and SCL)	analog inputs 4 and 5	PC4 and PC5
SPI	inaccessible to user	

## Pin Assignment Table Sorted by Pin

Arduino Pin	Orangutan Function	Notes/Alternate Functions
digital pin 0	digital I/O	USART input pin (RXD)
digital pin 1	digital I/O	<b>connected to red user LED</b> (high turns LED on) USART output pin (TXD)
digital pin 2	<b>LCD control line RS</b>	external interrupt 0 (INT0)
digital pin 3	<b>M2 control line</b>	Timer2 PWM output B (OC2B)
digital pin 4	<b>LCD control line E</b>	USART external clock input/output (XCK) Timer0 external counter (T0)
digital pin 5	<b>M1 control line</b>	Timer0 PWM output B (OC0B)
digital pin 6	<b>M1 control line</b>	Timer0 PWM output A (OC0A)
digital pin 7	<b>LCD data line DB7</b>	<b>connected to green user LED</b> (high turns LED on)
digital pin 8	<b>LCD control line R/W</b>	Timer1 input capture (ICP1) divided system clock output (CLK0)
digital pin 9	<b>LCD data line DB4</b>	<b>user pushbutton</b> (pressing pulls pin low) Timer1 PWM output A (OC1A)
digital pin 10	<b>buzzer</b>	Timer1 PWM output B (OC1B)
digital pin 11	<b>M2 control line</b>	Timer2 PWM output A (OC2A) ISP programming line
digital pin 12	<b>LCD data line DB5</b>	<b>user pushbutton</b> (pressing pulls pin low) <b>Caution: also an ISP programming line</b>
digital pin 13	<b>LCD data line DB6</b>	<b>user pushbutton</b> (pressing pulls pin low) <b>Caution: also an ISP programming line</b>
analog input 0	analog input and digital I/O	ADC input channel 0 (ADC0)
analog input 1	analog input and digital I/O	ADC input channel 1 (ADC1)
analog input 2	analog input and digital I/O	ADC input channel 2 (ADC2)
analog input 3	analog input and digital I/O	ADC input channel 3 (ADC3)
analog input 4	analog input and digital I/O	ADC input channel 4 (ADC4) I2C/TWI input/output data line (SDA)
analog input 5	analog input and digital I/O	ADC input channel 5 (ADC5) I2C/TWI clock line (SCL)
analog input 6	dedicated analog input	<b>SMT-jumpered to temperature sensor</b> (LV-168 only) <b>SMT-jumpered to battery voltage monitor</b> (SV-168 only) ADC input channel 6 (ADC6)

analog input 7	dedicated analog input	<b>jumpered to user trimmer potentiometer</b>
		ADC input channel 7 (ADC7)
reset	reset pushbutton	internally pulled high; active low
		digital I/O disabled by default

## 2.b. Baby Orangutan B Pin Mappings




## Pin Assignment Table Sorted by Function

Function	Arduino Pin	mega168 Pin
digital I/Os (x16)	digital pins 0 – 2, 4, 7 – 10, 12, 13 analog inputs 0 – 5	PD0 – PD2, PD4, PD7, PB0 – PB2, PB4, PB5, PC0 – PC5
analog inputs (x8)	analog inputs 0 – 7	PC0 – PC5, ADC6, ADC7
motor 1 control (A and B)	digital pins 5 and 6	PD5 and PD6
motor 2 control (A and B)	digital pins 3 and 11	PD3 and PB3
red user LED	digital pin 1	PD1
user trimmer potentiometer	analog input 7	ADC7
ICSP programming lines (x3)	digital pins 11, 12, and 13	PB3, PB4, PB5
$\overline{\text{RESET}}$	reset	PC6
UART (RX and TX)	digital pins 0 and 1	PD0 and PD1
I2C/TWI (SDA and SCL)	analog inputs 4 and 5	PC4 and PC5
SPI	inaccessable to user	
Timer1 PWM outputs (A and B)	digital pins 9 and 10	PB1 and PB2


## Pin Assignment Table Sorted by Pin

Arduino Pin	Orangutan Function	Notes/Alternate Functions
digital pin 0	digital I/O	USART input pin (RXD)
digital pin 1	digital I/O	<b>connected to red user LED</b> (high turns LED on) USART output pin (TXD)
digital pin 2	digital I/O	external interrupt 0 (INT0)
digital pin 3	<b>M2 control line</b>	Timer2 PWM output B (OC2B)
digital pin 4	digital I/O	USART external clock input/output (XCK) Timer0 external counter (T0)
digital pin 5	<b>M1 control line</b>	Timer0 PWM output B (OC0B)
digital pin 6	<b>M1 control line</b>	Timer0 PWM output A (OC0A)
digital pin 7	digital I/O	
digital pin 8	digital I/O	Timer1 input capture (ICP1) divided system clock output (CLK0)
digital pin 9	digital I/O	Timer1 PWM output A (OC1A)
digital pin 10	digital I/O	Timer1 PWM output B (OC1B)
digital pin 11	<b>M2 control line</b>	Timer2 PWM output A (OC2A) ISP programming line
digital pin 12	digital I/O	<b>Caution: also an ISP programming line</b>
digital pin 13	digital I/O	<b>Caution: also an ISP programming line</b>
analog input 0	analog input and digital I/O	ADC input channel 0 (ADC0)
analog input 1	analog input and digital I/O	ADC input channel 1 (ADC1)
analog input 2	analog input and digital I/O	ADC input channel 2 (ADC2)
analog input 3	analog input and digital I/O	ADC input channel 3 (ADC3)
analog input 4	analog input and digital I/O	ADC input channel 4 (ADC4) I2C/TWI input/output data line (SDA)
analog input 5	analog input and digital I/O	ADC input channel 5 (ADC5) I2C/TWI clock line (SCL)
analog input 6	dedicated analog input	ADC input channel 6 (ADC6)
analog input 7	dedicated analog input	<b>connected to user trimmer potentiometer</b> ADC input channel 7 (ADC7)
reset	$\overline{\text{RESET}}$ pin	internally pulled high; active low digital I/O disabled by default


## 2.c. 3pi Robot Pin Mappings



General features of the Pololu 3pi robot, top view.



Labeled bottom view of the Pololu 3pi robot.



Specific features of the Pololu 3pi robot, top view.

## Pin Assignment Table Sorted by Function

Function	Arduino Pin	mega168 Pin
free digital I/Os (x3) (remove PC5 jumper to free digital pin 19)	digital pins 0, 1, 19	PD0, PD1, PC5
free analog inputs (if you remove jumpers, x3)	analog inputs 5 – 7	PC5, ADC6, ADC7
motor 1 (left motor) control (A and B)	digital pins 5 and 6	PD5 and PD6
motor 2 (right motor) control (A and B)	digital pins 3 and 11	PD3 and PB3
QTR-RC reflectance sensors (left to right, x5)	digital pins 14 – 18	PC0 – PC4
red (left) user LED	digital pin 1	PD1
green (right) user LED	digital pin 7	PD7
user pushbuttons (left to right, x3)	digital inputs 9, 12, and 13	PB1, PB4, and PB5
buzzer	digital pin 10	PB2
LCD control (RS, R/W, E)	digital pins 2, 8, and 4	PD2, PB0, and PD4
LCD data (4-bit: DB4 – DB7)	digital pins 9, 12, 13, and 7	PB1, PB4, PB5, and PD7
reflectance sensor IR LED control (drive low to turn IR LEDs off)	digital pin 19 (through jumper)	PC5
user trimmer potentiometer	analog input 7 (through jumper)	ADC7
2/3rds of battery voltage	analog input 6 (through jumper)	ADC6
ICSP programming lines (x3)	digital pins 11, 12, and 13	PB3, PB4, PB5
reset pushbutton	reset	PC6
UART (RX and TX)	digital pins 0 and 1	PD0 and PD1
I2C/TWI	inaccessable to user	
SPI	inaccessable to user	

## Pin Assignment Table Sorted by Pin

Arduino Pin	Orangutan Function	Notes/Alternate Functions
digital pin 0	free digital I/O	USART input pin (RXD)
digital pin 1	free digital I/O	<b>connected to red user LED</b> (high turns LED on) USART output pin (TXD)
digital pin 2	<b>LCD control line RS</b>	external interrupt 0 (INT0)
digital pin 3	<b>M2 control line</b>	Timer2 PWM output B (OC2B)
digital pin 4	<b>LCD control line E</b>	USART external clock input/output (XCK) Timer0 external counter (T0)
digital pin 5	<b>M1 control line</b>	Timer0 PWM output B (OC0B)
digital pin 6	<b>M1 control line</b>	Timer0 PWM output A (OC0A)
digital pin 7	<b>LCD data line DB7</b>	<b>connected to green user LED</b> (high turns LED on)
digital pin 8	<b>LCD control line R/W</b>	Timer1 input capture (ICP1) divided system clock output (CLK0)
digital pin 9	<b>LCD data line DB4</b>	<b>user pushbutton</b> (pressing pulls pin low) Timer1 PWM output A (OC1A)
digital pin 10	<b>buzzer</b>	Timer1 PWM output B (OC1B)
digital pin 11	<b>M2 control line</b>	Timer2 PWM output A (OC2A) ISP programming line
digital pin 12	<b>LCD data line DB5</b>	<b>user pushbutton</b> (pressing pulls pin low) <b>Caution: also an ISP programming line</b>
digital pin 13	<b>LCD data line DB6</b>	<b>user pushbutton</b> (pressing pulls pin low) <b>Caution: also an ISP programming line</b>
digital pin 14	<b>QTR-RC reflectance sensor</b>	(drive high for 10 us, then wait for line input to go low) sensor labeled PC0 (leftmost sensor)
digital pin 15	<b>QTR-RC reflectance sensor</b>	(drive high for 10 us, then wait for line input to go low) sensor labeled PC1
digital pin 16	<b>QTR-RC reflectance sensor</b>	(drive high for 10 us, then wait for line input to go low) sensor labeled PC2 (center sensor)
digital pin 17	<b>QTR-RC reflectance sensor</b>	(drive high for 10 us, then wait for line input to go low) sensor labeled PC3
digital pin 18	<b>QTR-RC reflectance sensor</b>	(drive high for 10 us, then wait for line input to go low) sensor labeled PC4 (rightmost sensor)
digital pin 19	analog input and digital I/O	<b>jumpered to sensors' IR LEDs</b> (driving low turns off emitters)

		ADC input channel 5 (ADC5)
analog input 6	dedicated analog input	<b>jumpered to 2/3rds of battery voltage</b> ADC input channel 6 (ADC6)
analog input 7	dedicated analog input	<b>jumpered to user trimmer potentiometer</b> ADC input channel 7 (ADC7)
reset	reset pushbutton	internally pulled high; active low digital I/O disabled by default

### 3. Configuring the Arduino Environment

The configuration instructions in this section explain how to alter the Arduino IDE so that it programs your Orangutan or 3pi robot using the correct settings via our **USB AVR Programmer** [<https://www.pololu.com/product/1300>], and provides a series of libraries that can help you interface with onboard hardware on your device.

1. Go to <http://www.arduino.cc/en/Main/Software> [<http://www.arduino.cc/en/Main/Software>] and download the latest Arduino software for your platform (Arduino 1.6.11 as of this writing).
2. Download the **Pololu Orangutan and 3pi Robot add-on for the Arduino IDE** [<https://www.pololu.com/file/0J1229/libpololu-arduino-161018.zip>] (111k zip).
3. Copy the libpololu-arduino folder from the ZIP file into the “hardware” subfolder of your Arduino sketchbook location. The Arduino sketchbook location is typically in your Documents folder in a subfolder named “Arduino”. You can see the sketchbook location in the Arduino IDE Preferences dialog, which is available from the File menu.

For example, if you are using Windows and you have not changed the sketchbook location, the “libpololu-arduino” folder should be copied to:

**C:\Users\<<username>\Documents\Arduino\hardware\libpololu-arduino**

If the “Arduino” or “hardware” folders do not exist yet, you will have to create them.

After installing the add-on and restarting the Arduino IDE, you should see several new features inside the Arduino IDE:

- You should see entries for the Orangutans in the Board menu.
- You should see an entry for the Pololu USB AVR Programmer in the Programmer menu.
- If you select one of the Orangutan entries in the Board menu, then you should see several new entries in the “Examples” menu, like “OrangutanMotors” and “OrangutanLCD”.

If you do not see these new features, then the add-on was probably not installed correctly and you should retry the installation instructions above and restart the Arduino IDE.

Older versions of the download are provided here for compatibility with older Arduino environments. (Please note that some of these older downloads only provide the libraries, not the hardware add-on).

- **libpololu-arduino-150324.zip** [<https://www.pololu.com/file/0J868/libpololu-arduino-150324.zip>] (111k zip), tested with Arduino 1.6.1

- **libpololu-arduino-150218.zip** [<https://www.pololu.com/file/0J859/libpololu-arduino-150218.zip>] (111k zip), tested with Arduino 1.6.0
- **PololuArduinoLibraries-120914** [<https://www.pololu.com/file/0J581/PololuArduinoLibraries-120914.zip>] (107k zip), tested with Arduino 1.0.1
- **PololuArduinoLibraries-111221** [<https://www.pololu.com/file/0J521/PololuArduinoLibraries-111221.zip>] (107k zip), tested with Arduino 1.0
- **PololuArduinoLibraries-101215** [<https://www.pololu.com/file/0J425/PololuArduinoLibraries-101215.zip>] (107k zip), tested with Arduino 0021
- **PololuArduinoLibraries-080929** [<https://www.pololu.com/file/0J134/PololuArduinoLibraries-080929.zip>] (86k zip), tested with Arduino 0012
- **PololuArduinoLibraries-080826** [<https://www.pololu.com/file/0J127/PololuArduinoLibraries-080826.zip>] (80k zip), tested with Arduino 0011

## 4. Using the Arduino IDE: Blinking an LED

Now it's time to take the Arduino IDE for a spin. Our first goal will be to load a simple LED-blinking sketch (Arduino program) onto our Orangutan or 3pi.

**Open the Arduino IDE and create the following sketch:**


```

1  /*
2  * Blink
3  *
4  * The basic Arduino example, modified to work for Orangutans.
5  * Turns on an LED for one second, then off for one second,
6  * and so on... We use pin 1 because Orangutans have an LED
7  * on PD1, which corresponds to Arduino pin 1.
8  *
9  * http://www.arduino.cc/en/Tutorial/Blink
10 * */
11
12 int ledPin = 1;           // LED connected to digital pin 1 (PD1) on Orangutans
13
14 void setup()             // run once, when the sketch starts
15 {
16   pinMode(ledPin, OUTPUT); // sets the digital pin as output
17 }
18
19 void loop()              // run over and over again
20 {
21   digitalWrite(ledPin, HIGH); // sets the LED on
22   delay(1000);              // waits for a second
23   digitalWrite(ledPin, LOW);  // sets the LED off
24   delay(1000);              // waits for a second
25 }


```

Copy the above code into a new, blank sketch.

**Select one of our specially-created boards using the Tools > Board menu:**




## Select the programmer:




If you are using a Pololu USB AVR Programmer, select “Pololu USB AVR Programmer” in the Programmer menu. Otherwise, consult the documentation of your programmer to find out which option to select.

## Select the programmer’s serial port:



If you are using a Pololu USB AVR Programmer, make sure you have installed the proper drivers (see the **programmer’s user’s guide** [<https://www.pololu.com/docs/0J36>]) and have connected it to your computer. Once this is done, you should be able to see its COM port listed under the **Tools > Serial Port** menu. If you are using Windows and do not know which COM port your programmer is on, you can determine this by bringing up your computer’s Device Manager and expanding the **Ports (COM & LPT)** list. The programmer provides two virtual serial ports, and the correct port to use is the one named “Pololu USB AVR Programmer Programming Port”.




## Program your Orangutan or 3pi with the sketch:

Make sure your Orangutan or 3pi is connected to power and turned on. The power LED on the board should be on.

Connect your programmer to the Orangutan or 3pi via its six-pin ISP header, making sure that pin 1 of the programmer lines up with pin 1 of the target’s programming header. The Orangutan SV-xx8, Orangutan LV-168, and 3pi each have a shrouded header that will enforce correct pin orientation, but the Baby Orangutan B does not. Pin 1 on the Baby Orangutan is marked by a copper arrow on the PCB and pin 1 on our USB AVR Programmer is marked by a raised plastic triangle on the IDC connector.

With the programmer connected to both your computer and your Orangutan or 3pi, click the **Upload** button (circled in the screen capture below) or press Ctrl+U to compile the sketch and upload it to your target device. The status bar at the bottom of the window will read “Done uploading.” when the upload is complete. If everything has worked correctly, you should now see the target’s red user LED blinking on and off every second. Please note that if you are using a 3pi and have not yet soldered in the optional through-hole LEDs, you will need to look at the underside of the robot to see the flashing LED.



## 5. Arduino Libraries for the Orangutan and 3pi Robot



**Note:** Because the Orangutans were not designed as official Arduino boards, and run at 20 MHz instead of 16 Mhz, not all existing Arduino library code will be directly compatible with them. Also, Orangutan-specific libraries might conflict with existing Arduino libraries if the two are competing for the same hardware peripherals (e.g. two libraries that try to use Timer1 in different ways will not be compatible).

## Overview

This section provides a series of Orangutan-specific libraries that will allow you to easily interface with the Orangutan/3pi hardware from the Arduino environment. It is important to note that the Baby Orangutan B is a hardware subset of the larger Orangutan SV-xx8 and LV-168, and these Orangutans are (almost) a hardware subset of the 3pi robot. What this means is that any library code that works on the Baby Orangutan B will also work on the Orangutan and 3pi robot, and (almost) any library code that works on the Orangutan will also work on the 3pi. The following subsections contain detailed documentation of each Orangutan Arduino library. **All of the libraries that apply to the Orangutan will work for the corresponding hardware on the 3pi robot.**


The only library class for which it is meaningful to have multiple instances is PololuQTRSensors. All the other classes consist entirely of static methods, so it does not make sense to instantiate them multiple times. The only library classes that need to be explicitly initialized before they are used are PololuQTRSensors and Pololu3pi, since these objects are initialized with user-specified parameters. None of the remaining Orangutan library objects needs to be explicitly initialized in your `setup()` function as initialization is automatically performed when needed.

As of version 101215, the C versions of the library functions are also available for use in your Arduino programs. This means that there are three possible ways for you to call library methods:

```
1 unsigned int x;
2
3 // C++ function through instantiated object
4 OrangutanAnalog analog; // this line is usually placed above setup()
5 x = analog.readTrimpot();
6
7 // C++ function through class scope
8 x = OrangutanAnalog::readTrimpot();
9
10 // C function
11 x = read_trimpot();
```

## Usage examples

Most libraries come with at least one example sketch that demonstrates how the library can be used. To load an example sketch, open the Arduino IDE and go to **File > Examples > Orangutan\_\_\_\_\_**.



The Pololu3pi library comes with several detailed line-following and maze-solving example sketches that can be loaded using **File > Examples > Pololu3pi**.

If you cannot find these examples in the Examples menu, make sure that you have followed the installation instructions in **Section 3** and try restarting the Arduino IDE.

### 5.a. OrangutanAnalog - Analog Input Library

## Overview

This library provides a set of methods that can be used to read analog voltage inputs, as well as functions specifically designed to read the value of the trimmer potentiometer (on the **3pi robot** [<https://www.pololu.com/product/975>], **Orangutan SV-xx8** [<https://www.pololu.com/product/1227>], **Orangutan LV-168** [<https://www.pololu.com/product/775>], and **Baby Orangutan B** [<https://www.pololu.com/product/1220>]), the battery voltage level in millivolts (3pi robot, SV-xx8), and the value of the temperature sensor in tenths of a degree F or C (on the Orangutan LV-168 only). This library gives you more control than existing Arduino analog input functions.

You do not need to initialize your OrangutanAnalog object before use. All initialization is performed automatically when needed.

All of the methods in this class are static; you should never have more than one instance of an OrangutanAnalog object in your sketch.

## OrangutanAnalog Methods

Complete documentation of this library's methods can be found in **Section 2** of the **Pololu AVR Library Command Reference** [<https://www.pololu.com/docs/0J18>].

## Usage Examples

This library comes with two example sketches that you can load by going to **File > Examples > OrangutanAnalog**. The example sketches that come with the **OrangutanMotors** library also make limited use of this library.

### 1. OrangutanAnalogExample

Demonstrates how you can use the methods in this library to read the analog voltage of the trimmer potentiometer in the background while the rest of your code executes. If the ADC is free, the program starts a conversion on the **TRIMPOT** analog input (channel 7), and then it proceeds to execute the rest of the code in `loop()` while the ADC hardware works. Polling of the `isConverting()` method allows the program to determine when the conversion is complete and to update its notion of the trimpot value accordingly. Feedback is given via the red user LED, whose brightness is made to scale with the trimpot position.

```
1  #include <OrangutanLEDS.h>
2  #include <OrangutanAnalog.h>
3
4  /*
5   * OrangutanAnalogExample for the 3pi, Orangutan SV-xx8,
6   *   Orangutan LV-168, or Baby Orangutan B
7   *
8   * This sketch uses the OrangutanAnalog library to read the voltage output
9   * of the trimpot in the background while the rest of the main loop executes.
10  * The LED is flashed so that its brightness appears proportional to the
11  * trimpot position.
12  *
13  * http://www.pololu.com/docs/0J17/5.a
14  * http://www.pololu.com
15  * http://forum.pololu.com
16  */
17
18  OrangutanLEDS leds;
19  OrangutanAnalog analog;
20
21  unsigned int sum;
22  unsigned int avg;
23  unsigned char samples;
24
25  void setup()                // run once, when the sketch starts
26  {
27    analog.setMode(MODE_8_BIT); // 8-bit analog-to-digital conversions
28    sum = 0;
29    samples = 0;
30    avg = 0;
31    analog.startConversion(TRIMPOT); // start initial conversion
32  }
33
34  void loop()                 // run over and over again
35  {
36    if (!analog.isConverting()) // if conversion is done...
37    {
38      sum += analog.conversionResult(); // get result
39      analog.startConversion(TRIMPOT); // and start next conversion
40      if (++samples == 20)
41      {
42        avg = sum / 20; // compute 20-sample average of ADC result
43        samples = 0;
44        sum = 0;
45      }
46    }
47
48    // when avg == 0, the red LED is almost totally off
49    // when avg == 255, the red LED is almost totally on
50    // brightness should scale approximately linearly in between
51    leds.red(LOW); // red LED off
52    delayMicroseconds(256 - avg);
53    leds.red(HIGH); // red LED on
54    delayMicroseconds(avg + 1);
55  }
```

## 2. OrangutanAnalogExample2

Intended for use on the Orangutan LV-168. Note that it will run on the 3pi robot and Orangutan SV-xx8, but the displayed temperature will be incorrect as the analog input connected to the temperature

sensor on the Orangutan LV-168 is connected to 2/3rds of the battery voltage on the 3pi and to 1/3rd of the battery voltage on the Orangutan SV-xx8. It displays on the LCD the trimmer potentiometer output in millivolts and the temperature sensor output in degrees Fahrenheit. If you hold a finger on the underside of the Orangutan LV-168's PCB near the center of the board, you should see the temperature reading slowly start to rise. Be careful not to zap the board with electrostatic discharge if you try this!

```
1 #include <OrangutanLCD.h>
2 #include <OrangutanAnalog.h>
3
4 /*
5  * OrangutanAnalogExample2: for the Orangutan LV-168
6  *
7  * This sketch uses the OrangutanAnalog library to read the voltage output
8  * of the trimpot (in millivolts) and to read the Orangutan LV-168's
9  * temperature sensor in degrees Farenheit. These values are printed to
10 * the LCD 10 times per second. This example is intended for use with the
11 * Orangutan LV-168, though all but the temperature-measuring portion
12 * will work on the 3pi robot (on the 3pi, analog input 6 connects to 2/3rds
13 * of the battery voltage rather than a temperature sensor) and the
14 * Orangutan SV-xx8 (on the SV-xx8, analog input 6 connects to 1/3rd of
15 * the battery voltage).
16 *
17 * You should see the trimpot voltage change as you turn it, and you can
18 * get the temperature reading to slowly increase by holding a finger on the
19 * underside of the Orangutan LV-168's PCB near the center of the board.
20 * Be careful not to zap the board with electrostatic discharge if you
21 * try this!
22 */
23
24 OrangutanLCD lcd;
25 OrangutanAnalog analog;
26
27 void setup() // run once, when the sketch starts
28 {
29   analog.setMode(MODE_10_BIT); // 10-bit analog-to-digital conversions
30 }
31
32 void loop() // run over and over again
33 {
34   lcd.gotoXY(0,0); // LCD cursor to home position (upper-left)
35   lcd.print(analog.toMillivolts(analog.readTripot())); // trimpot output in mV
36   lcd.print(" mV "); // added spaces are to overwrite left over chars
37
38   lcd.gotoXY(0, 1); // LCD cursor to start of the second line
39
40   // get temperature in tenths of a degree F
41   unsigned int temp = analog.readTemperatureF();
42   lcd.print(temp/10); // get the whole number of degrees
43   lcd.print('.'); // print the decimal point
44   lcd.print(temp - (temp/10)*10); // print the tenths digit
45   lcd.print((char)223); // print a degree symbol
46   lcd.print("F "); // added spaces are to overwrite left over chars
47
48   delay(100); // wait for 100 ms (reduces LCD flicker)
49 }
```

## 5.b. OrangutanBuzzer - High-Level Buzzer Control Library

## Overview

This library allows you to easily control the piezo buzzer on the **3pi robot** [<https://www.pololu.com/product/975>], **Orangutan SV-xx8** [<https://www.pololu.com/product/1227>], and **Orangutan LV-168** [<https://www.pololu.com/product/775>]. It gives you the option of playing either a note or a frequency for a specified duration at a specified volume, or you can use the **play()** method to play an entire melody in the background. Buzzer control is achieved using one of the Timer1 PWM outputs, and duration timing is performed using a Timer1 overflow interrupt, so **this library will conflict with any other libraries that rely on or reconfigure Timer1**. For example, the Arduino function **analogWrite()** will not work on the Timer1 PWM output pins once you have started to use this library in your sketch.

The benefit to this approach is that you can play notes on the buzzer while leaving the CPU mostly free to execute the rest of your code. This means you can have a melody playing in the background while your Orangutan or 3pi does its main task. You can poll the **isPlaying()** method to determine when the buzzer is finished playing.

You do not need to initialize your OrangutanBuzzer object before use. All initialization is performed automatically when needed.

All of the methods in this class are static; you should never have more than one instance of an OrangutanBuzzer object in your sketch.

## OrangutanBuzzer Methods

Complete documentation of this library's methods can be found in **Section 3** of the **Pololu AVR Library Command Reference** [<https://www.pololu.com/docs/0J18>].

## Usage Examples

This library comes with three example sketches that you can load by going to **File > Examples > OrangutanBuzzer**.

### 1. OrangutanBuzzerExample

Demonstrates one way to use this library's **playNote()** method to play a simple melody stored in RAM. It should immediately start playing the melody, and you can use the top user pushbutton to stop and replay the melody. The example is structured so that you can add your own code to **loop()** and the melody will still play normally in the background, assuming your code executes quickly enough to avoid inserting delays between the notes. You can use this same technique to play melodies that have been stored in EEPROM (the mega168 has enough room in EEPROM to store 170 notes).

```

1  #include <OrangutanLCD.h>
2  #include <OrangutanPushbuttons.h>
3  #include <OrangutanBuzzer.h>
4
5  /*
6   * OrangutanBuzzerExample: for the Orangutan SV-xx8, Orangutan LV-168,
7   *   and 3pi robot
8   *
9   * This example uses the OrangutanBuzzer library to play a series of notes on
10  * the buzzer. It also uses the OrangutanLCD library to display the notes it is
11  * playing, and it uses the OrangutanPushbuttons library to allow the user to
12  * stop/reset the melody with the top pushbutton.
13  *
14  * http://www.pololu.com/docs/0J17/5.b
15  * http://www.pololu.com
16  * http://forum.pololu.com
17  */
18
19 #define MELODY_LENGTH 95
20
21 // These arrays take up a total of 285 bytes of RAM (out of a 1k limit)
22 unsigned char note[MELODY_LENGTH] =
23 {
24     NOTE_E(5), SILENT_NOTE, NOTE_E(5), SILENT_NOTE,
25     NOTE_E(5), SILENT_NOTE, NOTE_C(5), NOTE_E(5),
26     NOTE_G(5), SILENT_NOTE, NOTE_G(4), SILENT_NOTE,
27
28     NOTE_C(5), NOTE_G(4), SILENT_NOTE, NOTE_E(4), NOTE_A(4),
29     NOTE_B(4), NOTE_B_FLAT(4), NOTE_A(4), NOTE_G(4),
30     NOTE_E(5), NOTE_G(5), NOTE_A(5), NOTE_F(5), NOTE_G(5),
31     SILENT_NOTE, NOTE_E(5), NOTE_C(5), NOTE_D(5), NOTE_B(4),
32
33     NOTE_C(5), NOTE_G(4), SILENT_NOTE, NOTE_E(4), NOTE_A(4),
34     NOTE_B(4), NOTE_B_FLAT(4), NOTE_A(4), NOTE_G(4),
35     NOTE_E(5), NOTE_G(5), NOTE_A(5), NOTE_F(5), NOTE_G(5),
36     SILENT_NOTE, NOTE_E(5), NOTE_C(5), NOTE_D(5), NOTE_B(4),
37
38     SILENT_NOTE, NOTE_G(5), NOTE_F_SHARP(5), NOTE_F(5),
39     NOTE_D_SHARP(5), NOTE_E(5), SILENT_NOTE, NOTE_G_SHARP(4),
40     NOTE_A(4), NOTE_C(5), SILENT_NOTE, NOTE_A(4), NOTE_C(5), NOTE_D(5),
41
42     SILENT_NOTE, NOTE_G(5), NOTE_F_SHARP(5), NOTE_F(5),
43     NOTE_D_SHARP(5), NOTE_E(5), SILENT_NOTE,
44     NOTE_C(6), SILENT_NOTE, NOTE_C(6), SILENT_NOTE, NOTE_C(6),
45
46     SILENT_NOTE, NOTE_G(5), NOTE_F_SHARP(5), NOTE_F(5),
47     NOTE_D_SHARP(5), NOTE_E(5), SILENT_NOTE,
48     NOTE_G_SHARP(4), NOTE_A(4), NOTE_C(5), SILENT_NOTE,
49     NOTE_A(4), NOTE_C(5), NOTE_D(5),
50
51     SILENT_NOTE, NOTE_E_FLAT(5), SILENT_NOTE, NOTE_D(5), NOTE_C(5)
52 };
53
54 unsigned int duration[MELODY_LENGTH] =
55 {
56     100, 25, 125, 125, 125, 125, 125, 125, 250, 250, 250, 250, 250,
57
58     375, 125, 250, 375, 250, 250, 125, 250, 167, 167, 167, 250, 125, 125,
59     125, 250, 125, 125, 375,
60
61     375, 125, 250, 375, 250, 250, 125, 250, 167, 167, 167, 250, 125, 125,
62     125, 250, 125, 125, 375,

```

```
63
64     250, 125, 125, 125, 250, 125, 125, 125, 125, 125, 125, 125, 125,
65
66     250, 125, 125, 125, 250, 125, 125, 200, 50, 100, 25, 500,
67
68     250, 125, 125, 125, 250, 125, 125, 125, 125, 125, 125, 125, 125,
69
70     250, 250, 125, 375, 500
71 };
72
73 OrangutanLCD lcd;
74 OrangutanPushbuttons buttons;
75 OrangutanBuzzer buzzer;
76 unsigned char currentIdx;
77
78 void setup()                // run once, when the sketch starts
79 {
80     currentIdx = 0;
81     lcd.print("Music!");
82 }
83
84 void loop()                 // run over and over again
85 {
86     // if we haven't finished playing the song and
87     // the buzzer is ready for the next note, play the next note
88     if (currentIdx < MELODY_LENGTH && !buzzer.isPlaying())
89     {
90         // play note at max volume
91         buzzer.playNote(note[currentIdx], duration[currentIdx], 15);
92
93         // optional LCD feedback (for fun)
94         lcd.gotoXY(0, 1); // go to start of the second LCD line
95         lcd.print((unsigned int)note[currentIdx]); // print integer value of current note
96         lcd.print(" "); // overwrite any left over characters
97         currentIdx++;
98     }
99
100    // Insert some other useful code here...
101    // the melody will play normally while the rest of your code executes
102    // as long as it executes quickly enough to keep from inserting delays
103    // between the notes.
104
105    // For example, let the top user pushbutton function as a stop/reset melody button
106    if (buttons.isPressed(TOP_BUTTON))
107    {
108        buzzer.stopPlaying(); // silence the buzzer
109        if (currentIdx < MELODY_LENGTH)
110            currentIdx = MELODY_LENGTH; // terminate the melody
111        else
112            currentIdx = 0; // restart the melody
113        buttons.waitForRelease(TOP_BUTTON); // wait here for the button to be released
114    }
115 }
```

## 2. OrangutanBuzzerExample2

Demonstrates how you can use this library's **play()** method to start a melody playing. Once started, the melody will play all the way to the end with no further action required from your code, and the rest of your program will execute as normal while the melody plays in the background. The **play()** method

is driven entirely by the Timer1 overflow interrupt. The top user pushbutton will play a fugue by Bach from program memory, the middle user pushbutton will quietly play the C major scale up and back down from RAM, and the bottom user pushbutton will stop any melody that is currently playing or play a single note if the buzzer is currently inactive.

```

1  #include <OrangutanLCD.h>
2  #include <OrangutanPushbuttons.h>
3  #include <OrangutanBuzzer.h>
4
5  /*
6   * OrangutanBuzzerExample2: for the Orangutan SV-xx8, Orangutan LV-168,
7   *   and 3pi robot
8   *
9   * This example uses the OrangutanBuzzer library to play a series of notes on
10  * the buzzer. It uses the OrangutanPushbuttons library to allow the user
11  * select which melody plays.
12  *
13  * This example demonstrates the use of the OrangutanBuzzer::play() method,
14  * which plays the specified melody entirely in the background, requiring
15  * no further action from the user once the method is called. The CPU
16  * is then free to execute other code while the melody plays.
17  *
18  * http://www.pololu.com/docs/0J17/5.b
19  * http://www.pololu.com
20  * http://forum.pololu.com
21  */
22
23  OrangutanLCD lcd;
24  OrangutanPushbuttons buttons;
25  OrangutanBuzzer buzzer;
26
27  #include <avr/pgmspace.h> // this lets us refer to data in program space (i.e. flash)
28  // store this fugue in program space using the PROGMEM macro.
29  // Later we will play it directly from program space, bypassing the need to load it
30  // all into RAM first.
31  const char fugue[] PROGMEM =
32  " ! 05 L16 agafaea dac+adaea fa<aa<bac#a dac#adaea f"
33  "06 dcd<b-d<ad<g d<f+d<gd<ad<b- d<dd<ed<f+d<g d<f+d<gd<ad"
34  "L8 MS <b-d<b-d MLe-<ge-<g MSc<ac<a ML d<fd<f 05 MS b-gb-g"
35  "ML >c#e>c#e MS afaf ML gc#g#c# MS fdfd ML e<b-e<b-"
36  "06 L16ragafaea dac#adaea fa<aa<bac#a dac#adaea faeadaca"
37  "<b-acadg<b-g egdgcg<b-g <ag<b-gcf<af dfcf<b-f<af"
38  "<gf<af<b-e<ge c#e<b-e<ae<ge <fe<ge<ad<fd"
39  "05 e>e>ef>df>d b->c#b->c#a>df>d e>e>ef>df>d"
40  "e>d>c#>db>d>c#b >c#agaegfe f 06 dc#dfdc<b c#4";
41
42
43  void setup() // run once, when the sketch starts
44  {
45    lcd.print("Press a");
46    lcd.gotoXY(0, 1);
47    lcd.print("button..");
48  }
49
50  void loop() // run over and over again
51  {
52    // wait here for one of the three buttons to be pushed
53    unsigned char button = buttons.waitForButton(ALL_BUTTONS);
54    lcd.clear();
55
56    if (button == TOP_BUTTON)
57    {
58      buzzer.playFromProgramSpace(fugue);
59
60      lcd.print("Fugue!");
61      lcd.gotoXY(0, 1);
62      lcd.print("flash ->");

```

```
63     }
64     if (button == MIDDLE_BUTTON)
65     {
66         buzzer.play("! V8 cdefgab>cbagfedc");
67         lcd.print("C Major");
68         lcd.gotoXY(0, 1);
69         lcd.print("RAM ->");
70     }
71     if (button == BOTTOM_BUTTON)
72     {
73         if (buzzer.isPlaying())
74         {
75             buzzer.stopPlaying();
76             lcd.print("stopped");
77         }
78         else
79         {
80             buzzer.playNote(NOTE_A(5), 200, 15);
81             lcd.print("note A5");
82         }
83     }
84 }
```

### 3. OrangutanBuzzerExample3

Demonstrates the use of this library's `playMode()` and `playCheck()` methods. In this example, automatic play mode is used to allow the melody to keep playing while it blinks the red user LED. Then the mode is switched to play-check mode during a phase where we are trying to accurately measure time. There are three `#define` macros that allow you to run this example in different ways and observe the result. Please see the comments at the top of the sketch for more detailed information.

```

1  #include <OrangutanLCD.h>
2  #include <OrangutanLEDs.h>
3  #include <OrangutanBuzzer.h>
4
5  /*
6   * OrangutanBuzzerExample3: for the Orangutan LV-168, Orangutan SV-xx8,
7   *   or 3pi robot
8   *
9   * This example uses the OrangutanBuzzer library to play a series of notes on
10  * the target's piezo buzzer.
11  *
12  * This example demonstrates the use of the OrangutanBuzzer::playMode()
13  * and OrangutanBuzzer::playCheck() methods, which allow you to select
14  * whether the melody sequence initiated by OrangutanBuzzer::play() is
15  * played automatically in the background by the Timer1 interrupt, or if
16  * the play is driven by the playCheck() method in your main loop.
17  *
18  * Automatic play mode should be used if your code has a lot of delays
19  * and is not time critical. In this example, automatic mode is used
20  * to allow the melody to keep playing while we blink the red user LED.
21  *
22  * Play-check mode should be used during parts of your code that are
23  * time critical. In automatic mode, the Timer1 interrupt is very slow
24  * when it loads the next note, and this can delay the execution of your.
25  * Using play-check mode allows you to control when the next note is
26  * loaded so that it doesn't occur in the middle of some time-sensitive
27  * measurement. In our example we use play-check mode to keep the melody
28  * going while performing timing measurements using Timer2. After the
29  * measurements, the maximum time measured is displayed on the LCD.
30  *
31  * Immediately below are three #define statements that allow you to alter
32  * the way this program runs. You should have one of the three lines
33  * uncommented while commenting out the other two:
34  *
35  * If only WORKING_CORRECTLY is uncommented, the program should run in its
36  * ideal state, using automatic play mode during the LED-blinking phase
37  * and using play-check mode during the timing phase. The maximum recorded
38  * time should be 20, as expected.
39  *
40  * If only ALWAYS_AUTOMATIC is uncommented, the program will use automatic
41  * play mode during both the LED-blinking phase and the timing phase. Here
42  * you will see the effect this has on the time measurements (instead of 20,
43  * you should see a maximum reading of around 27 or 28).
44  *
45  * If only ALWAYS_CHECK is uncommented, the program will be in play-check
46  * mode during both the LED-blinking phase and the timing phase. Here you
47  * will see the effect that the LED-blinking delays have on play-check
48  * mode (the sequence will be very choppy while the LED is blinking, but
49  * sound normal during the timing phase). The maximum timing reading should
50  * be 20, as expected.
51  */
52
53  // *** UNCOMMENT ONE OF THE FOLLOWING PRECOMPILER DIRECTIVES ***
54  // (the remaining two should be commented out)
55  #define WORKING_CORRECTLY // this is the right way to use playMode()
56  //#define ALWAYS_AUTOMATIC // playMode() is always PLAY_AUTOMATIC (timing is inaccurate)
57  //#define ALWAYS_CHECK // playMode() is always PLAY_CHECK (delays interrupt the sequence)
58
59  OrangutanLEDs leds;
60  OrangutanBuzzer buzzer;
61  OrangutanLCD lcd;
62

```

```

63 #include <avr/pgmspace.h>
64 const char rhapsody[] PROGMEM = "06 T40 L16 d#<b<f#<d#<f#<bd#f#"
65 "T80 c#<b-<f#<c#<f#<b-c#8"
66 "T180 d#b<f#d#f#>bd#f#c#b-<f#c#f#>b-c#8 c>c#<c#>c#<b>c#<c#>c#<c#>c#<b>c#<c#>c#"
67 "c>c#<c#>c#<b->c#<c#>c#<c#>c#<b->c#<c#>c#"
68 "c>c#<c#>c#f>c#<c#>c#<c#>c#f>c#<c#>c#"
69 "c>c#<c#>c#f#>c#<c#>c#<c#>c#f#>c#<c#>c#d#bb-bd#bf#d#c#b-ab-c#b-f#d#";
70
71
72 void setup() // run once, when the sketch starts
73 {
74   TCCR2A = 0; // configure timer2 to run at 78 kHz
75   TCCR2B = 0x06; // and overflow when TCNT2 = 256 (~3 ms)
76   buzzer.playFromProgramSpace(rhapsody);
77 }
78
79 void loop() // run over and over again
80 {
81   // allow the sequence to keep playing automatically through the following delays
82 #ifndef ALWAYS_CHECK
83   buzzer.playMode(PLAY_AUTOMATIC);
84 #else
85   buzzer.playMode(PLAY_CHECK);
86 #endif
87   lcd.gotoXY(0, 0);
88   lcd.print("blink!");
89   int i;
90   for (i = 0; i < 8; i++)
91   {
92 #ifdef ALWAYS_CHECK
93     buzzer.playCheck();
94 #endif
95     leds.red(HIGH);
96     delay(500);
97     leds.red(LOW);
98     delay(500);
99   }
100
101   lcd.gotoXY(0, 0);
102   lcd.print("timing");
103   lcd.gotoXY(0, 1);
104   lcd.print(" "); // clear bottom LCD line
105   // turn off automatic playing so that our time-critical code won't be interrupted by
106   // the buzzer's long timer1 interrupt. Otherwise, this interrupt could throw off our
107   // timing measurements. Instead, we will now use playCheck() to keep the sequence
108   // playing in a way that won't throw off our measurements.
109 #ifndef ALWAYS_AUTOMATIC
110   buzzer.playMode(PLAY_CHECK);
111 #endif
112   unsigned char maxTime = 0;
113   for (i = 0; i < 8000; i++)
114   {
115     TCNT2 = 0;
116     while (TCNT2 < 20) // time for ~250 us
117     ;
118     if (TCNT2 > maxTime)
119       maxTime = TCNT2; // if the elapsed time is greater than the previous max, save it
120 #ifndef ALWAYS_AUTOMATIC
121     buzzer.playCheck(); // check if it's time to play the next note and play it if so
122 #endif
123   }
124   lcd.gotoXY(0, 1);

```

```
125 | lcd.print("max=");  
126 | lcd.print((unsigned int)maxTime);  
127 | lcd.print(' '); // overwrite any left over characters  
128 | }
```

### 5.c. OrangutanLCD - LCD Control Library

## Overview

This library gives you the ability to control the 8×2 character LCD on the **3pi robot** [<https://www.pololu.com/product/975>], **Orangutan SV-xx8** [<https://www.pololu.com/product/1227>], and **Orangutan LV-168** [<https://www.pololu.com/product/775>]. It implements the standard 4-bit HD44780 protocol, and it uses the busy-wait-flag feature to avoid the unnecessarily long delays present in other 4-bit LCD Arduino libraries. This comprehensive library is meant to offer as much LCD control as possible, so it most likely gives you more methods than you need. You can comment out unneeded methods (e.g. `showCursor()`) in `OrangutanLCD.cpp` to make this library smaller. If you do this, remember to delete `OrangutanLCD.o` so and restart the Arduino IDE so that the library will be recompiled.

This library is designed to gracefully handle alternate use of the four LCD data lines. It will change their data direction registers and output states only when needed for an LCD command, after which it will immediately restore the registers to their previous states. This allows the LCD data lines to function as pushbutton inputs and an LED driver on the Orangutan and 3pi.

You do not need to initialize your `OrangutanLCD` object before use. All initialization is performed automatically when needed.

All of the methods in this class are static; you should never have more than one instance of an `OrangutanLCD` object in your sketch.

## OrangutanLCD Methods

Complete documentation of this library's methods can be found in **Section 5** of the **Pololu AVR Library Command Reference** [<https://www.pololu.com/docs/0J18>].

## Usage Examples

This library comes with two example sketches that you can load by going to **File > Examples > OrangutanLCD**. Note that most of the other libraries have example sketches that use the LCD, so please see these for more OrangutanLCD usage examples.


### 1. OrangutanLCDExample

Demonstrates shifting the contents of the display by moving the word “Hello” around the two lines of the LCD.

```
1  #include <OrangutanLCD.h>
2
3  /*
4  * OrangutanLCDExample for the Orangutan LV-168, Orangutan SV-xx8,
5  *   or 3pi robot
6  *
7  * This example uses the OrangutanLCD library to write "Hello"
8  * on the LCD and then move it around the display area.
9  */
10
11 OrangutanLCD lcd;
12
13 void setup()           // run once, when the sketch starts
14 {
15 }
16
17 void loop()           // run over and over again
18 {
19   lcd.print("Hello"); // display "Hello" at (0, 0), a.k.a. upper-left
20   delay(200);
21
22   // shift the display right every 200ms three times
23   lcd.scroll(LCD_RIGHT, 3, 200);
24   lcd.clear();        // clear the LCD
25   lcd.gotoXY(3, 1);   // go to the fourth character of the second LCD line
26   lcd.print("Hello"); // display "Hello" at (3, 1), a.k.a. lower-right
27   delay(200);
28
29   // shift the display left every 200ms three times
30   lcd.scroll(LCD_LEFT, 3, 200);
31   lcd.clear();        // clear the LCD
32 }
```

### 1. OrangutanLCDExample2

Demonstrates creating and displaying custom characters on the LCD. The following picture shows an example of custom characters, using them to display a bar graph of sensor readings and a smiley face:



```
1  #include <OrangutanPushbuttons.h>
2  #include <OrangutanLCD.h>
3  #include <stdlib.h>          // used for its "random" and "srandom" functions
4
5  /*
6   * OrangutanLCDExample2: for the Orangutan SV-xx8, Orangutan LV-168 or 3pi robot
7   *
8   * This example uses the OrangutanLCD library to display custom characters on
9   * the LCD.  Simply push a any user pushbutton to display a new, randomly
10  * chosen mood character.
11  */
12
13  OrangutanLCD lcd;
14
15  // define some custom "mood" characters
16  #include <avr/pgmspace.h> // this lets us refer to data in program space (i.e. flash)
17  const char happy[] PROGMEM = {
18    0b00000, // the 5 bits that make up the top row of the 5x8 character
19    0b01010,
20    0b01010,
21    0b01010,
22    0b00000,
23    0b10001,
24    0b01110,
25    0b00000
26  };
27
28  const char sad[] PROGMEM = {
29    0b00000,
30    0b01010,
31    0b01010,
32    0b01010,
33    0b00000,
34    0b01110,
35    0b10001,
36    0b00000
37  };
38
39  const char indifferent[] PROGMEM = {
40    0b00000,
41    0b01010,
42    0b01010,
43    0b01010,
44    0b00000,
45    0b00000,
46    0b01110,
47    0b00000
48  };
49
50  const char surprised[] PROGMEM = {
51    0b00000,
52    0b01010,
53    0b01010,
54    0b00000,
55    0b01110,
56    0b10001,
57    0b10001,
58    0b01110
59  };
60
61  const char mocking[] PROGMEM = {
62    0b00000,
```

```
63     0b01010,
64     0b01010,
65     0b01010,
66     0b00000,
67     0b11111,
68     0b00101,
69     0b00010
70 };
71
72 char prevMood = 5;
73
74 void setup()                // run once, when the sketch starts
75 {
76     lcd.loadCustomCharacter(happy, 0);
77     lcd.loadCustomCharacter(sad, 1);
78     lcd.loadCustomCharacter(indifferent, 2);
79     lcd.loadCustomCharacter(surprised, 3);
80     lcd.loadCustomCharacter(mocking, 4);
81     lcd.clear();           // must be called before we can use the custom chars
82     lcd.print("mood: ?");
83
84     // initialize the random number generator based on how long we hold the button the first time
85     OrangutanPushbuttons::waitForPress(ALL_BUTTONS);
86     long seed = 0;
87     while(OrangutanPushbuttons::isPressed(ALL_BUTTONS))
88         seed++;
89     srandom(seed);         // the same as: randomSeed((unsigned int)seed);
90 }
91
92 void loop()                // run over and over again
93 {
94     lcd.gotoXY(6, 0);     // move cursor to the correct position
95
96     // ensure we get a new mood that differs from the previous
97     char mood;
98     do
99     {
100         mood = random()%5; // the same as: mood = random(5);
101     } while (mood == prevMood);
102     prevMood = mood;
103
104     lcd.print(mood);     // print a random mood character
105
106     // wait for any button to be pressed
107     OrangutanPushbuttons::waitForButton(ALL_BUTTONS);
108 }
```

## 5.d. OrangutanLEDs - LED Control Library

## Overview

This library allows you to easily control the LED(s) on the **3pi robot** [<https://www.pololu.com/product/975>], **Orangutan SV-xx8** [<https://www.pololu.com/product/1227>], **Orangutan LV-168** [<https://www.pololu.com/product/775>], and **Baby Orangutan B** [<https://www.pololu.com/product/1220>]. On the Orangutan SV-xx8 and LV-168, there are two user LEDs on the top side of the PCB with the red LED on the bottom left and the green LED on the top right. On the 3pi, there are two user LEDs on the bottom side of the PCB with the red LED on the right (when looking at the bottom) and the green LED on the left. Additional LEDs included with the 3pi may be soldered in on the top side (in parallel with the surface-mount LEDs on the underside) for easier viewing. The Baby Orangutan has a single red LED and no green LED.

Note that the red LED is on the same pin as the UART0 serial transmitter (PD1), so if you are using UART0 for serial transmission then the red LED commands will not work, and you will see the red LED blink briefly whenever data is transmitted on UART0. Note that the green LED is on the same pin as an LCD control pin; the green LED will blink briefly whenever data is sent to the LCD, but the two functions will otherwise not interfere with each other.

You do not need to initialize your OrangutanLEDs object before use. All initialization is performed automatically when needed.

All of the methods in this class are static; you should never have more than one instance of an OrangutanLEDs object in your sketch.

## OrangutanLEDs Methods

Complete documentation of this library's methods can be found in **Section 6** of the **Pololu AVR Library Command Reference** [<https://www.pololu.com/docs/0J18>].

## Usage Examples

This library comes with an example sketch that you can load by going to **File > Examples > OrangutanLEDs**.

### 1. OrangutanLEDExample

Alternately blinks the red and green LEDs on the Orangutan SV-xx8, Orangutan LV-168, or 3pi robot. If you run this program on the Baby Orangutan B, you will only see the red user LED blink, but you can connect an external LED to pin PD7 (Arduino pin 7) if you want to see the second LED blink. If you do this, don't forget to include a current-limiting resistor!

```
1  #include <OrangutanLEDs.h>
2
3  /*
4   * OrangutanLEDExample: for the 3pi robot, Orangutan LV 168, Orangutan SV-xx8,
5   *   or Baby Orangutan B.
6   *
7   * This program uses the OrangutanLEDs functions to control the red and green
8   * LEDs on the 3pi robot or Orangutan. It will also work to control the red
9   * LED on the Baby Orangutan B (which lacks a second, green LED).
10  */
11
12  OrangutanLEDs leds;
13
14  void setup()                // run once, when the sketch starts
15  {
16  }
17
18  void loop()                 // run over and over again
19  {
20    leds.red(HIGH);           // red LED on
21    delay(1000);              // waits for a second
22    leds.red(LOW);            // red LED off
23    delay(1000);              // waits for a second
24    leds.green(HIGH);         // green LED on (will not work on the Baby Orangutan)
25    delay(500);               // waits for 0.5 seconds
26    leds.green(LOW);          // green LED off (will not work on the Baby Orangutan)
27    delay(500);               // waits for 0.5 seconds
28  }
```

### 5.e. OrangutanMotors - Motor Control Library

## Overview

This library gives you the ability to control the motor drivers on the **3pi robot** [<https://www.pololu.com/product/975>], **Orangutan SV-xx8** [<https://www.pololu.com/product/1227>], **Orangutan LV-168** [<https://www.pololu.com/product/775>], and **Baby Orangutan B** [<https://www.pololu.com/product/1220>]. It accomplishes this by using the four hardware PWM outputs from timers Timer0 and Timer2, so **this library will conflict with any other libraries that rely on or reconfigure Timer0 or Timer2.**



Because the Arduino environment relies on Timer0 for timing functions like `millis()` and `delay()`, this library cannot reconfigure the timers; consequently, **the PWM outputs are limited to a frequency of 1.25 kHz in the Arduino environment.** In many cases, this will result in an audible motor whine. To avoid this problem, you can instead program in C or C++ with the **Pololu AVR Library** [<https://www.pololu.com/docs/0J20>], which allows you to use the motor drivers at higher PWM frequencies.

You do not need to initialize your `OrangutanMotors` object before use. All initialization is performed automatically when needed.

All of the methods in this class are static; you should never have more than one instance of an `OrangutanMotors` object in your sketch.

## OrangutanMotors Methods

Complete documentation of this library's methods can be found in **Section 7** of the **Pololu AVR Library Command Reference** [<https://www.pololu.com/docs/0J18>].

## Usage Examples

This library comes with two example sketches that you can load by going to **File > Examples > OrangutanMotors**.

### 1. OrangutanMotorExample

Demonstrates controlling the motors using the trimmer potentiometer and uses the red LED for feedback.

```
1 #include <OrangutanLEDs.h>
2 #include <OrangutanAnalog.h>
3 #include <OrangutanMotors.h>
4
5
6 /*
7  * OrangutanMotorExample for the 3pi robot, Orangutan LV-168, Orangutan SV-xx8,
8  *   and Baby Orangutan B
9  *
10 * This example uses the OrangutanMotors library to drive
11 * motors in response to the position of user trimmer potentiometer
12 * and blinks the red user LED at a rate determined by the trimmer
13 * potentiometer position. It uses the OrangutanAnalog library to measure
14 * the trimpot position, and it uses the OrangutanLEDs library to provide
15 * limited feedback with the red user LED.
16 *
17 * http://www.pololu.com/docs/0J17/5.e
18 * http://www.pololu.com
19 * http://forum.pololu.com
20 */
21
22 OrangutanAnalog analog;
23 OrangutanLEDs leds;
24 OrangutanMotors motors;
25
26 void setup()           // run once, when the sketch starts
27 {
28
29 }
30
31 void loop()           // run over and over again
32 {
33 // note that the following line could also be accomplished with:
34 // int pot = analog.read(7);
35 int pot = analog.readTrimpot(); // determine the trimpot position
36 int motorSpeed = pot/2-256; // turn pot reading into number between -256 and 255
37 if(motorSpeed == -256)
38     motorSpeed = -255; // 256 is out of range
39 motors.setSpeeds(motorSpeed, motorSpeed);
40
41 int ledDelay = motorSpeed;
42 if(ledDelay < 0)
43     ledDelay = -ledDelay; // make the delay a non-negative number
44 ledDelay = 256-ledDelay; // the delay should be short when the speed is high
45
46 leds.red(HIGH); // turn red LED on
47 delay(ledDelay);
48
49 leds.red(LOW); // turn red LED off
50 delay(ledDelay);
51 }
```

## 2. OrangutanMotorExample2

Demonstrates controlling the motors using the trimmer potentiometer, but it uses the LCD for most of the feedback, so it will not fully work on the Baby Orangutan.

```
1  #include <OrangutanLEDs.h>
2  #include <OrangutanAnalog.h>
3  #include <OrangutanMotors.h>
4  #include <OrangutanLCD.h>
5
6  /*
7   * OrangutanMotorExample2 for the 3pi robot, Orangutan LV-168,
8   *   and Orangutan SV-xx8.
9   *
10  * This example uses the OrangutanMotors and OrangutanLCD libraries to drive
11  * motors in response to the position of user trimmer potentiometer
12  * and to display the potentiometer position and desired motor speed
13  * on the LCD. It uses the OrangutanAnalog library to measure the
14  * trimpot position, and it uses the OrangutanLEDs library to provide
15  * limited feedback with the red and green user LEDs.
16  */
17
18  OrangutanLCD lcd;
19  OrangutanMotors motors;
20  OrangutanAnalog analog;
21  OrangutanLEDs leds;
22
23  void setup()                // run once, when the sketch starts
24  {
25
26  }
27
28  void loop()                 // run over and over again
29  {
30  // note that the following line could also be accomplished with:
31  // int pot = analogRead(7);
32  int pot = analog.readTripot(); // determine the trimpot position
33
34  // avoid clearing the LCD to reduce flicker
35  lcd.gotoXY(0, 0);
36  lcd.print("pot=");
37  lcd.print(pot);           // print the trim pot position (0 - 1023)
38  lcd.print(" ");          // overwrite any left over digits
39
40  int motorSpeed = (512 - pot) / 2;
41  lcd.gotoXY(0, 1);
42  lcd.print("spd=");
43  lcd.print(motorSpeed);    // print the resulting motor speed (-255 - 255)
44  lcd.print(" ");
45  motors.setSpeeds(motorSpeed, motorSpeed); // set speeds of motors 1 and 2
46
47
48  // all LEDs off
49  leds.red(LOW);
50  leds.green(LOW);
51  // turn green LED on when motors are spinning forward
52  if (motorSpeed > 0)
53    leds.green(HIGH);
54  // turn red LED on when motors are spinning in reverse
55  if (motorSpeed < 0)
56    leds.red(HIGH);
57  delay(100);
58  }
```

## 5.f. OrangutanPushbuttons - Pushbutton Interface Library

## Overview

This library allows you to easily interface with the three user pushbuttons on the **3pi robot** [<https://www.pololu.com/product/975>], **Orangutan SV-xx8** [<https://www.pololu.com/product/1227>], and **Orangutan LV-168** [<https://www.pololu.com/product/775>] by either polling for the state of specific buttons or by waiting for press/release events on specifiable buttons. The **waitFor\_\_\_\_\_()** methods in this library automatically take care of button debouncing.

You do not need to initialize your `OrangutanPushbuttons` object before use. All initialization is performed automatically when needed.

All of the methods in this class are static; you should never have more than one instance of an `OrangutanPushbuttons` object in your sketch.

## OrangutanPushbuttons Methods

Complete documentation of this library's methods can be found in **Section 9** of the **Pololu AVR Library Command Reference** [<https://www.pololu.com/docs/0J18>].

## Usage Examples

This library comes with an example sketch that you can load by going to **File > Examples > OrangutanPushbuttons**.

### 1. OrangutanPushbuttonExample

Demonstrates interfacing with the user pushbuttons. It will wait for you to push either the top button or the bottom button, at which point it will display on the LCD which button was pressed. It will also detect when that button is subsequently released and display that to the LCD.

```
1  #include <OrangutanLCD.h>
2  #include <OrangutanPushbuttons.h>
3
4  /*
5   * OrangutanPushbuttonExample: for the 3pi robot, Orangutan LV-168,
6   *   and Orangutan SV-xx8
7   *
8   * This example uses the OrangutanPushbuttons library to detect user input
9   * from the pushbuttons, and it uses the OrangutanLCD library to display
10  * feedback on the LCD.
11  */
12
13  OrangutanPushbuttons buttons;
14  OrangutanLCD lcd;
15
16  void setup()                // run once, when the sketch starts
17  {
18  }
19
20  void loop()                 // run over and over again
21  {
22    lcd.clear();
23    lcd.print("Waiting");
24
25    // wait for either the top or bottom buttons to be pressed
26    // store the value of the pressed button in the variable 'button'
27    unsigned char button = buttons.waitForPress(TOP_BUTTON | BOTTOM_BUTTON);
28    lcd.clear();
29    if (button == TOP_BUTTON) // display the button that was pressed
30      lcd.print("top down");
31    else
32      lcd.print("bot down");
33    buttons.waitForRelease(button); // wait for that button to be released
34    lcd.clear();
35    lcd.print("released");     // display that the button was released
36    delay(1000);
37  }
```

### 5.g. Pololu3pi - Sensor Library for the 3pi Robot

## Overview

This library allows you to easily interface with the five infrared reflectance sensors on the **3pi robot** [<https://www.pololu.com/product/975>]. Note that in order to use this library, **you must also include PololuQTRSensors.h in your sketch**. You should have something like the following at the top of your sketch:

```
1 #include <Pololu3pi.h> // gives access to sensor interface functions
2 #include <PololuQTRSensors.h> // used by Pololu3pi.h
3 #include <OrangutanMotors.h> // gives access to motor control functions
4 #include <OrangutanBuzzer.h> // gives access to buzzer control functions
```

Unlike the other Orangutan libraries, you must explicitly call the **init()** method to initialize your Pololu3pi object before using it.



All of the methods in this class are static; you should never have more than one instance of a Pololu3pi object in your sketch.

## Pololu3pi Methods

Complete documentation of this library's methods can be found in **Section 19** of the **Pololu AVR Library Command Reference** [<https://www.pololu.com/docs/0J18>].

## Usage Examples

This library comes with three example sketches that you can load by going to **File > Examples > Pololu3pi**. There is a simple line-following example, a more advanced PID-based line-following example, and a basic maze-solving example.



For more information on creating line and maze courses, as well as a detailed explanation of the C versions of the demo programs, please see sections 6 and 7 of the **3pi User's Guide** [<https://www.pololu.com/docs/0J21>].